

1.9 Lekce 8 – Práce se sešitem

Kromě manipulace buněk v rámci listů je v Excelu možné přistupovat k celému sešitu. V programovacím jazyce VBA je manipulování celý sešitem primárně řešeno objekty *Worksheets* a *Sheets*, o kterých bude primární řeč v této lekci.

1.9.1 Vybrání konkrétního listu sešitu

Přistoupit ke konkrétnímu objektu listu sešitu lze pomocí objektu *Sheets*. Vývojář může jako argument vložit index listu nebo celý název listu jako řetězec. Vybírání sešitu probíhá pomocí funkce *Select*, která po zavolání zobrazí vybraný list uživateli v Excelu.

```
' použití názvu listu  
Sheets("NovyList").Select  
  
' použití indexu listu  
Sheets(2).Select
```

Obrázek 72 – Ukázka kódu objektu *Sheets* a funkce *Select*

Objekt *Sheets* lze procházet jako pole pomocí *For Each* cyklu, například jde zobrazit nebo vypsat všechny názvy listů v sešitě pomocí vlastnosti *Name*. Objekt *Sheets* zahrnuje i listy s grafy, na rozdíl od objektu *Worksheets*.

```
' Vypíše názvy všech listů v sešitě  
For Each Sh In Sheets  
    MsgBox (Sh.Name)  
Next Sh
```

Obrázek 73 – Ukázka procházení objektu *Sheets*

<poznámka ohledně aktivování kódu pomocí *Activate*>

1.9.2 Zamykání a odemykání listů

V objektu *Sheets* je dostupná metoda *Protect*, která uzamkne vybraný list sešitu. Metoda má několik nepovinných parametrů. Nejdůležitějším parametrem je *password*, který uzamkne sešit společně se zadaným heslem. Ostatní parametry nastavují možnosti práce s listem, jak jej po zamknutí lze upravovat.

```
' Zápís zamknutí pouze s heslem
Sheets(1).Protect ("heslo")
' Alternativně
Sheets(1).Protect Password:="heslo"

' Ostatní parametry funkce Protect
' DrawingObjects, Contents, Scenarios, UserInterfaceOnly,
' AllowFormattingCells, AllowFormattingColumns, AllowFormattingRows,
' AllowInsertingColumns, AllowInsertingRows, AllowInsertingHyperlinks,
' AllowDeletingColumns, AllowDeletingRows, AllowSorting, AllowFiltering,
' AllowUsingPivotTables
```

Obrázek 74 – Ukázka kódu uzamknutí listu

Odemykání listu probíhá podobnou procedurou *Unprotect*, která má volitelný parametr *Password* se stejnou funkcí jako u procedury *Protect*.

```
' Odemknutí listu sešitu
Sheets(1).Unprotect ("heslo")
```

Obrázek 75 – Ukázka kódu odemknutí sešitu

1.9.3 Zobrazení listů

Listy lze zobrazit a skrýt pomocí vlastnosti objektu *Sheets* pojmenované *Visible*. Pokud má vlastnost hodnotu *PRAVDA*, list je zobrazený a viditelný v sešitě. Naopak s hodnotou *NEPRAVDA* je list skryt.

```
Sub SkrytNovyList()

' Skryje list, lze obnovit z Excelu
Sheets("NovyList").Visible = xlSheetHidden

' Skryje list, nelze obnovit z Excelu
Sheets("NovyList").Visible = xlSheetVeryHidden

' Zviditelní list
Sheets("NovyList").Visible = xlSheetVisible

End Sub
```

Obrázek 76 – Ukázka kódu skrytí a zobrazení sešitu v listě

Alternativně lze použít předdefinované hodnoty *xlSheetHidden*, *xlSheetVeryHidden* a *xlSheetVisible*. Hodnota *xlSheetVeryHidden* nastaví list skrytý bez možnosti obnovit jeho viditelnost uživatelem v Excelu. Vždy je nutné takový list zviditelnit pomocí makra.

1.9.4 Přidání a odebrání listů

Přidání listů do sešitu probíhá pomocí metody *Add*, která může obsahovat dva parametry o nové pozici listu. Parametr *Before* určuje pozici před zmíněným sešitem, kde se nový sešit vloží. Parametr *After* zase pozici po určeném sešitě, kde se nový sešit objeví.

```
' Přidá nový list na začátek sešitu
Worksheets.Add

' Přidá nový list před událostí
Worksheets.Add Before:=Worksheets("Události")

' Přidá nový list za událostí
Worksheets.Add After:=Worksheets("Události")
```

Obrázek 77 – Ukázka kódu přidání listu do sešitu

Odstranění listu probíhá jeho výběrem a zavoláním metody *Delete*. Smazaný list a jeho data nelze zpětně obnovit. Při odstraňování se Excel zeptá, zda chcete sešit opravdu odstranit.

```
' Odebere list
Worksheets("Přidaný list").Delete
```

Obrázek 78 – Ukázka kódu odebrání listu ze sešitu

1.9.5 Kopírování a přesunování listů

V rámci objektu *Worksheets* je dostupná metoda *Copy*, která bez uvedeného parametru zkopíruje vybraný list aktivního sešitu a vloží jej do sešitu nového. Zkopírovaný list v novém sešitě bude mít stejný název jako měl předtím.

```
' Zkopíruje sešit do nového sešitu
Sheets("NovyList").Copy

' Zkopíruje list NovyList za NovyList
Worksheets("NovyList").Copy After:=Worksheets("NovyList")

' Zkopíruje list NovyList před NovyList
Worksheets("NovyList").Copy Before:=Worksheets("NovyList")
```

Obrázek 79 – Ukázka kódu kopírování listu v sešitě

Vývojář ovšem ve většině případů nepotřebuje nakopírovat listy do nového sešitu. Metoda *Copy* má z tohoto důvodu dva parametry *After* a *Before*. Parametry označují nové místo pro kopírovaný list sešitu.

Pokud je potřeba novou kopii sešitu přejmenovat, stačí upravit vlastnost *Name* objektu *ActiveSheet* novým řetězcem s požadovaným názvem. Po kopírování by vždy měla nastat situace, že se zkopírovaný list zobrazí a stane se aktivním (objektem *ActiveSheet*).

```
' Označení více listů
Worksheets("Události").Select
Worksheets("NovyList").Select False

' Kopírování vybraných listů
ActiveWindow.SelectedSheets.Copy
```

Obrázek 80 – Ukázka kódu s kopírováním více listů zároveň

Pro kopii více sešitů zároveň, existuje výběr *SelectedSheets* v objektu *ActiveWindow*, který odkazuje na všechny právě vybrané listy. Označení více listů probíhá pomocí *Select* s přidaným parametrem *False*, díky kterému se neodznačí předchozí vybrané listy.

1.9.6 Události v listu

V sešitě mohou existovat předdefinovaná makra, která se automaticky spustí při nějaké akci spuštěné uživatelem (události). Akcí může být relativně cokoliv, od jednoduché změny hodnoty v listu až po dvoj kliknutí na buňku v listě.

```
Private Sub Worksheet_SelectionChange(ByVal Target As Range)
' Nastala změna v listu Události ve výběru
MsgBox ("Nastala událost výběru v listu! Nový výběr bude označen oranžově. ")


' Zruší barvu předchozího výběru
Sheets("Události").Cells.Interior.ColorIndex = xlNone

' Označí novou oblast barvou
Target.Interior.Color = RGB(252, 186, 3)

End Sub
```

Obrázek 81 – Ukázka kódu události *SelectionChange*

Všechn kód metod událostí musí být v objektu listu, a nikoliv Modulu jako u předchozích případech. Ve vybraném objektu se vybere objekt *Worksheet* a z druhé nabídky procedur událost, pro kterou je potřeba vytvořit kód, který se vždy při vzniku takové události spustí.



Obrázek 82 – Ukázka přidání makra události ke konkrétnímu listu

Jednotlivé typy událostí je vhodné řádně prostudovat v rámci dokumentace VBA, např. na adrese <https://learn.microsoft.com/en-us/office/vba/excel/concepts/events-worksheetfunctions-shapes/worksheet-object-events>.

```

Dim pocet kliknuti As Integer

Private Sub Worksheet_BeforeDoubleClick(ByVal Target As Range, Cancel As Boolean)
    If (pocet_kliknuti Mod 2 = 0) Then
        Target = Target.Value & "X"
    Else
        Target = Target.Value & "O"
    End If

    pocet_kliknuti = pocet_kliknuti + 1
End Sub

Private Sub Worksheet_SelectionChange(ByVal Target As Range)
    ' Nastala změna v listu Události ve výběr

```

Obrázek 83 – Ukázka kódu jiné události *BeforeDoubleClick*

1.9.7 Úkoly k procvičení

- 1) Vytvořte makro *Zamek*, které zamkne a odemkne list sešitu dle vašeho hesla
- 2) Vytvořte nové makro *NovaTabulka*, které založí nový list s hlavičkou tabulky: Typ paliva, množství (v l), cena celkem.
- 3) Pomocí dalšího makra *SkrytNoveMakro* skryjte nově vytvořený list.
- 4) Vytvořte kód, který při změně vybraného listu ohlásí název vybraného listu.